


Self-Biased BFP405

- SIEGET[®]25- Technology
- Small SCT598-Package
- Control Pin For Switching The Device Off
- Current Easy Adjustable By An External Resistor
- Voltage Independent Current (2V – 4.5V)




VPW05982


ESD: Electrostatic discharge sensitive device, observe handling precautions!

Type	Marking	Ordering Code (8-mm taped)	Pin Configuration (circuit Diagram)	Package
BGC405	40s	Q62702-G0091	see below	SCT598

Equivalent Circuit



Pin Connections, SCT598



Note: Top View

Description

The BGC405 is a silicon self biased RF Transistor (Q1). It offers an adjustable collector current nearly independent from device voltage in the range from 2.0V to 4.5V. Additionally a control pin (Vc) for switching the device off is provided. The collector current can be adjusted by connecting a resistor (Rx) between Vcc and Vr.

Maximum Ratings

Parameter	Symbol		Unit
Device current	I_{CC}	12	mA
Device voltage	V_{CC}	4.5	V
Total power dissipation, $T_s \leq 120^\circ\text{C}$ ¹⁾	P_{tot}	54	mW
Control voltage	V_C	$V_{CC}+0.5$	V
Input Current for pin 1	I_r	380	μA
Junction temperature	T_j	150	$^\circ\text{C}$
Ambient temperature range	T_A	-65...+150	$^\circ\text{C}$
Storage temperature range	T_{stg}	-65...+150	$^\circ\text{C}$

Thermal Resistance

Junction-soldering point ¹⁾	$R_{th JS}$	≤ 530	K/W
--	-------------	------------	-----

¹⁾ T_S is measured on the Ground lead at the soldering point to the pcb.


Electrical Specifications (Measured in Test Fixture applying the circuit specified in Figure 1 with $R_x=82\Omega$, $T_c=25^\circ\text{C}$, $V_{CC}=3\text{V}$, $I_{CC}\approx 7\text{mA}$ unless noted)

Symbol	Parameter	Unit	Min	Typ	Max
Gp	Power Gain ($ S_{21} ^2$)	f=900MHz	19.5	21	
		f=1.8GHz	16.5	18	
NF	Noise Figure (in 50 Ω System)	f=900MHz		1.8	2.1
		f=1.8GHz		2.0	2.3
P_{-1dB}	Output Power at 1dB Gain Compression (in 50 Ω System)	f=900MHz		1	
		f=1.8GHz		0.5	
IP_3	Third Order Intercept Point (Output, Γ_{Opt})	f=900MHz		15	
		f=1.8GHz		15	
RL_{in}	Input Return Loss	f=900MHz		5	
		f=1.8GHz		8	
RL_{out}	Output Return Loss	f=900MHz		1.5	
		f=1.8GHz		3	
t_{on}	On Switching Time ³⁾	μs		3.7	
t_{off}	Off Switching Time ³⁾	μs		2.5	
I_{leak}	Leakage Current In Sleep Mode	μA		<10	
I_{VcOn}	Control Pin (Vc) Current in Active Mode ²⁾	μA		35	
I_{VcOff}	Control Pin (Vc) Current in Sleep Mode ²⁾	nA		-60	
V_{cmin}	Minimum Voltage at Vc for Sleep Mode	V		$V_{CC} - 0.3\text{V}$	
V_{cmax}	Maximum Voltage at Vc for Active Mode	V		$0\text{V}+0.3\text{V}$	


²⁾ A positive sign denotes a current flowing from the Pin to the external circuit.

³⁾ These values are valid for $C_2=1\text{nF}$, $C_3=100\text{pF}$ and 220pF Coupling capacitors at R_{Fin} and R_{Fout} .


**Power Gain
versus Frequency**
 $V_{cc}=3V$, $I_{cc}=5mA$




**Power Gain
versus Device Current**
 $V_{cc}=3V$




**$|S_{21}|^2$
versus Frequency and Temperature**
 $V_{cc}=3V$, $I_{cc}=7mA$




**Device Current
versus Device Voltage**



**Device Current
versus R_x and Temperature
 $V_{cc}=3V$**



**Device Current
versus Voltage at V_c
 $V_{cc}=3V$; $R_x=82\Omega$**



Typical Application




Figure 1. Typical Application and Internal Circuit

Remarks:

- 1) To provide low frequency stability C2 should be 10 times C3.
- 2) Be aware that also coupling capacitors determine the switching times.
- 3) The collector current at Q1 can be estimated by $I_c = 0.6\text{V} / R_x[\Omega]$.
- 4) Place C2 as close to the device as possible.

Layout Proposal




Figure 2. Layout Proposal

Part List for $V_{CC}=3V$, $I_{CC}\approx 7mA$

Component	Value	Comment
L1	100nH	RFC
C2	1nF	Compensation Capacitor for Low Frequency Stabilization
C3	100pF	RFC
C4	150pF	Blocking Capacitor
C5	100nF	Blocking Capacitor
C6	220pF	Coupling Capacitor
C7	220pF	Coupling Capacitor
Rx	82Ω	Current Adjust
Substrate	h=0.5mm	Fr4, $\epsilon_r=4.5$
BGC405		

This proposal demonstrates how to use the BGC405 as a *Self-Biased Transistor*. As for a discrete Transistor matching circuits have to be applied. A good starting point for various applications are the Application Notes provided for the BFP405.

SPICE Model

The following SPICE Listing describes the circuit shown in figure 3. It is valid for low frequencies. For frequencies above 100MHz the parasitic circuit elements noted in figure 4 and table 1 should be added.




Figure. 3: Circuit used in the SPICE File

```
* Preliminary SPICE Model for BGC405
* SIEMENS HIGH FREQUENCY PRODUCTS
* Small Scale MMIC Design Group
```

```
.PARAM R=82
```

```
** Analysis setup **
```

```
*.TRAN 2ns 15u 0 2n
```

```
.TEMP +27 -40 +85
```

```
.DC LIN V1 0V 4V 0.1V
```

```
*.DC LIN V2 0V 3V 0.1V
```

```
*.STEP PARAM R LIST 33 47 68 100 150
```

```
* Voltage supply
```

```
V1 Vcc 0 DC 3.0V
```

```
V2 Vc 0 DC 0.0V
```

```
*Vpul Vc 0 PULSE(0 3V 100ns 0 0 9us 1000m)
```

* Internal Resistors

```
R1    3 Vc      47k    TC=-0.0006,0.0000025
R2    Vr 2      500    TC=-0.0006,0.0
R3    Vb 0      10k    TC=-0.0006,0.0000025
R4    Vb rfin   2.7k   TC=-0.0006,0.0
```

* External Resistors

```
Rx    Vcc Vr    {R}    TC=+0.000050,0.0
Rout  vout 0     50
Rin   vin 0     50
```

* External Capacitors

```
C2    Vb 0      1nF
C3    Vr 0      100pF
C7    rfin vin   220pF
C6    rfout vout 220pF
```

* Inductors (external)

```
L1    Vr rfout   100nH
```

* Transistors

```
Q1    rfout rfin 0      BFP405
X2    2 3 Vb 0          8PL18
X3    Vcc 5 5 0         2PL18
X4    5 3 3 0           2PL18
```

.PROBE

.MODEL BFP405 NPN(

```
+ IS = 1.9969e-16      BF = 83.23      NF = 1.0405
+ VAF = 39.251         IKF = 0.16493    ISE = 1.5761e-14
+ NE = 1.7763          BR = 10.526      NR = 0.96647
+ VAR = 34.368         IKR = 0.25052    ISC = 3.7223e-17
+ NC = 1.3152          RB = 15          IRB = 0.00021215
+ RBM = 1.3491         RE = 1.9289      RC = 0.12691
+ CJE = 3.7265e-15     VJE = 0.70367    MJE = 0.37747
+ TF = 4.5899e-12      XTF = 0.3641     VTF = 0.19762
+ ITF = 0.0013364      PTF = 0          CJC = 9.6941e-14
+ VJC = 0.99532        MJC = 0.48652    XCJC = 0.08161
+ TR = 1.4935e-09      CJS= 0           VJS = 0.75
+ MJS = 0              XTB = 0           EG = 1.11
+ XTI = 3              FC = 0.99469)
```

* PNP: PL18 E B C Bulk

.SUBCKT 8PL18 3 2 1 94

```
Q1    993    2    3    94    TL18 8
Q2    94     2    3    94    VSL18 8
Q3    94     2    993   94    LSL18 8
RCEX  993    1              0.204
.ENDS
```

.SUBCKT 2PL18 3 2 1 94


```

Q1    993    2    3    94    TL18  2
Q2    94     2    3    94    VSL18 2
Q3    94     2   993    94    LSL18 2
RCEX  993    1                0.816
.ENDS

```

```

.MODEL          TL18          PNP
+IS             = 2.914E-17    NF           = 1.000E+00    BF           = 4.005E+02
+NE             = 1.553E+00    ISE          = 6.923E-16    NR           = 1.000E+00
+BR             = 2.869E+01    NC           = 1.500E+00    ISC          = 8.190E-15
+VAF            = 6.000E+01    IKF          = 1.676E-04    VAR          = 2.214E+00
+IKR            = 2.474E-05    RB           = 6.000E+01    IRB          = 0.000E+00
+RBM            = 4.000E+01    RE           = 2.597E+00    RC           = 4.000E+00
+XTB            = -6.000E-01    EG           = 1.156E+00    XTI          = 3.000E+00
+CJE            = 1.200E-14    VJE          = 4.900E-01    MJE          = 1.360E-01
+TF             = 7.600E-10    XTF          = 2.872E-01    VTF          = 1.000E+03
+ITF            = 1.400E-02    CJC          = 4.700E-13    VJC          = 7.610E-01
+MJC            = 3.760E-01    XCJC         = 1.000E+00    TR           = 0.000E+00
+CJS            = 0.000E+00    VJS          = 7.500E-01    MJS          = 0.000E+00
+PTF            = 0.000E+00    FC           = 5.000E-01

```

```

.MODEL          VSL18         PNP
+IS             = 1.630E-19    NF           = 1.000E+00    BF           = 1.000E+09
+NE             = 1.500E+00    ISE          = 0.000E+00    NR           = 1.000E+00
+BR             = 1.000E+09    NC           = 2.000E+00    ISC          = 0.000E+00
+VAF            = 1.000E+02    IKF          = 1.794E-04    VAR          = 1.700E+00
+IKR            = 1.000E+00    RB           = 0.000E+00    IRB          = 0.000E+00
+RBM            = 0.000E+00    RE           = 0.000E+00    RC           = 0.000E+00
+XTB            = 0.000E+00    EG           = 1.122E+00    XTI          = 3.000E+00
+CJE            = 0.000E+00    VJE          = 6.800E-01    MJE          = 3.400E-01
+TF             = 2.000E-09    XTF          = 0.000E+00    VTF          = 1.000E+03
+ITF            = 1.000E+06    CJC          = 1.950E-13    VJC          = 5.500E-01
+MJC            = 3.770E-01    XCJC         = 0.000E+00    TR           = 0.000E+00
+CJS            = 0.000E+00    VJS          = 7.500E-01    MJS          = 0.000E+00
+PTF            = 0.000E+00    FC           = 5.000E-01

```

```

.MODEL          LSL18         PNP
+IS             = 4.261E-17    NF           = 1.000E+00    BF           = 1.000E+09
+NE             = 1.500E+00    ISE          = 0.000E+00    NR           = 1.000E+00
+BR             = 1.000E+09    NC           = 2.000E+00    ISC          = 0.000E+00
+VAF            = 6.000E+01    IKF          = 9.648E-05    VAR          = 1.700E+00
+IKR            = 1.000E+00    RB           = 0.000E+00    IRB          = 0.000E+00
+RBM            = 0.000E+00    RE           = 0.000E+00    RC           = 0.000E+00
+XTB            = 0.000E+00    EG           = 1.158E+00    XTI          = 3.000E+00
+CJE            = 0.000E+00    VJE          = 6.800E-01    MJE          = 3.400E-01
+TF             = 1.000E-09    XTF          = 0.000E+00    VTF          = 1.000E+03
+ITF            = 1.000E+06    CJC          = 0.000E+00    VJC          = 4.600E-01
+MJC            = 3.000E-01    XCJC         = 0.000E+00    TR           = 0.000E+00
+CJS            = 0.000E+00    VJS          = 7.500E-01    MJS          = 0.000E+00
+PTF            = 0.000E+00    FC           = 5.000E-01
*****

```

.END





Figure 4. Parasitic circuit elements for frequencies above 100MHz

Element	Value
Lp1	0.58nH
Lp2	0.56nH
Lp3	0.23nH
Lp4	0.05nH
Lp5	0.53nH
Lp6	0.47nH
Lp7	1nH
Cp1	134fF
Cp2	136fF
Cp3	6.9fF

Table 1. Parasitic circuit elements for frequencies above 100MHz

Package



Published by
Infineon Technologies AG i Gr.,
Bereichs Kommunikation
St.-Martin-Strasse 76,
D-81541 München
 © Infineon Technologies AG 1999
All Rights Reserved.

Attention please!

The information herein is given to describe certain components and shall not be considered as warranted characteristics.

Terms of delivery and rights to technical change reserved.

We hereby disclaim any and all warranties, including but not limited to warranties of non-infringement, regarding circuits, descriptions and charts stated herein.

Infineon Technologies is an approved CECC manufacturer.

Information

For further information on technology, delivery terms and conditions and prices please contact your nearest Infineon Technologies Office in Germany or our Infineon Technologies Representatives worldwide (see address list).

Warnings

Due to technical requirements components may contain dangerous substances. For information on the types in question please contact your nearest Infineon Technologies Office.

Infineon Technologies Components may only be used in life-support devices or systems with the express written approval of Infineon Technologies, if a failure of such components can reasonably be expected to cause the failure of that life-support device or system, or to affect the safety or effectiveness of that device or system. Life support devices or systems are intended to be implanted in the human body, or to support and/or maintain and sustain and/or protect human life. If they fail, it is reasonable to assume that the health of the user or other persons may be endangered.